

赤尾の “現場で勝てる” Visual Basic

Visual Basic 6.0ユーザーに贈る
イチから出直し超基本プログラミング

6.0

最終回

自分で作るActiveXコントロール

赤尾 猫太 AKAO, Nekota

その2

「コントロールの位置とサイズ」「OCXの作成」 「自作コントロールの配布」



ドライブ、フォルダ、 ファイルに関する処理

早速、前回作成した自作コントロール (FileSelect.ctl) の続きを作成しましょう。以前 (本誌2003年2月号) にもお話したとおり、ディレクトリリストボックスコントロール、ドライブリストボックスコントロール、ファイルリストボックスコントロールは勝手には連動してくれません。そこで以前と同様に、ディレクトリリストボックスコントロールとドライブリストボックスコントロールのChangeイベントに、ドライブやパスの移動に対して他のコントロールの対象を移動するためのコードを記述します (リスト1)。

続いて、ファイルリストボックスコントロールでファ

イルを選択したときのコードを記述します (リスト2)。これも以前と同様で、選択されたファイル名は配列であるListプロパティのListIndex番目に存在するのでそれを取り出します。しかし、Listプロパティにはファイル名しか入っていないので、Pathプロパティの中にあるパスと合わせてフルパス付きファイル名を生成して、SelectDataプロパティにセットするために代入を行なっています。

リスト1: コントロールの対象を移動するためのChangeイベント

```
Private Sub Dir1_Change()  
    'パスの移動を設定  
    File1.Path = Dir1.Path  
End Sub
```

```
Private Sub Drive1_Change()  
    'パスの移動を設定  
    Dir1.Path = Drive1.Drive  
End Sub
```

リスト2: 選択したファイル名をフルパス付きで取得する

```
Private Sub File1_Click()  
    'ファイルのクリックによりプロパティに  
    'フルパス付きファイル名をセット  
    Dim dt As String  
  
    dt = File1.Path  
    If Right$(dt, 1) <> "\" Then  
        dt = dt & "\"  
    End If  
    dt = dt & File1.List(File1.ListIndex)  
    SelectData = dt  
End Sub
```

本稿で前提となるもの

OS Windows 98以上


開発環境 Visual Basic 5.0/6.0



この記事で解説しているサンプルプログラムは、付録CD-ROMの¥DMAG¥AKAOフォルダ以下に収録しています。

¥FILESELECT : 自作コントロール
¥VIEWER1 : 以前に作成したイメージビューワ
¥VIEWER2 : 自作コントロールを組み込んだイメージビューワ





Resizeイベント

続いて、Resizeイベント (UserControl_Resize) です。コントロールも貼り付けたときや実行時に、サイズの変化 (リサイズ) が発生することがあるので、それに対処するためのコードを記述します (リスト3)。

まずは、必要な変数や定数を宣言します。ここでの定数は、作成しているコントロール上のドライブリストボックスやディレクトリリストボックスなど配置したコントロールの間隔を定数で“100twips”としているものです。間隔をもっとあげたいときなどはこの値を変更すればよいでしょうし、この設定をプロパティとして提供することで変更できるようにするのもよいでしょう。

次に、現在のコントロールのサイズを取り出します。以前、フォーム上に配置したコントロールをフォームのサイズに合わせて表示するために使ったのは、Formの

ScaleWidthプロパティやScaleHeightプロパティでした。しかし今回は、見てのとおり、コントロールの土台であるフォームのようなものには、タイトルバーもボーダーも存在していません。そこで今回は、UserControlのWidthプロパティとHeightプロパティを使用します。この値を基に、以降で各コントロールのサイズと位置の指定を行なっています。

続いて、サイズの下限を確認しています (リスト3-①)。この部分を記述せずに実行してコントロールを思いきり小さくしてみると、エラーになってしまいます。本連載で過去に作成したプログラムの中で、Resizeイベントに対応している処理を実行した際に同じようなエラーになったことがある方もいるでしょう。これらは、演算によりコントロールの大きさや位置などを指定しているため、大きさとしてありえない値になった際にエラーとなるのです。

もちろんエラーが発生しなくても、配置したそれぞれ

リスト3：UserControlのResizeイベント

```
Private Sub UserControl_Resize()
    ' サイズの変更に対応
    Dim x As Integer
    Dim y As Integer
    Const s As Integer = 100

    ' 現在のサイズ取り出し
    x = UserControl.Width
    y = UserControl.Height

    ' 異常発生を抑えるためサイズの下限を設定
    If x < 2000 Or y < 2000 Then
        If x < 2000 Then
            x = 2000
            UserControl.Width = 2000
        End If

        If y < 2000 Then
            y = 2000
            UserControl.Height = 2000
        End If
    End If

    ' 設定
    With Drive1
        .Top = s
        .Left = s
        .Width = (x - s * 3) / 2
    End With

    With Dir1
        .Top = Drive1.Height + s * 2
        .Left = s
        .Width = (x - s * 3) / 2
        .Height = y - Drive1.Height - Command1.Height - s * 4
    End With

    With File1
        .Top = s
        .Left = Drive1.Width + s * 2
        .Width = (x - s * 3) / 2
        .Height = y - Command1.Height - s * 3
    End With

    With Command1
        .Top = y - Command1.Height - s
        .Left = s
        .Width = (x - s * 3) / 2
    End With

    With Command2
        .Top = y - Command1.Height - s
        .Left = Command1.Width + s * 2
        .Width = (x - s * 3) / 2
    End With
End Sub
```

